
Abstract: This paper presents a computational algorithm to derive the theory of linear gauge invariant
perturbations on anisotropic cosmological spacetimes of the Bianchi I type. Our code is based on
the tensor algebra packages xTensor and xPert, within the computational infrastructure of xAct
written in Mathematica. The algorithm is based on a Hamiltonian, or phase space formulation, and
it provides an efficient and transparent way of isolating the gauge invariant degrees of freedom in
the perturbation fields and to obtain the Hamiltonian generating their dynamics. The restriction to
Friedmann–Lemaître–Robertson–Walker spacetimes is straightforward.

Keywords: primordial perturbations; anisotropic spacetime; Hamiltonian formulation

1. Introduction

Our understanding of the physics of the early universe is connected to cosmological perturbation
theory. This framework describes the evolution of cosmological spacetimes, together with matter
and gravitational perturbations propagating thereon. The description of perturbations is technically
complicated, mainly due to the fact that general relativity is a gauge theory. Due care is needed to
separate gauge artifacts from physical effects. A natural and conceptually clean strategy is to work
with gauge invariant fields, which are combinations of matter and gravitational perturbations that
are left invariant under diffeomorphisms, or coordinate transformations, at the desired order in the
perturbative expansion. This strategy was first implemented by Bardeen [1] by expanding Einstein’s
equations to first order in perturbations, and it has been extensively used since then. An alternative and
more geometric treatment can be obtained by working in the Hamiltonian, or phase space formulation
of general relativity [2]. Here, the phase space is equipped with four constraints—the so-called scalar
and vector constraints—which are the generators of gauge transformations. Working at leading
order in perturbations, one defines the gauge invariant fields as those combinations that remain
unchanged under the transformations generated by the linearized constraints or, in simpler words,
that Poisson-commute with them. Hence, at the practical level, the task of finding gauge invariant
perturbations reduces to finding combinations of linear perturbations whose Poisson brackets with
the linearized constraints vanish. Furthermore, this procedure is equivalent to finding an appropriate
canonical transformation, in such a way that the search for gauge invariant fields reduces to solving
a Hamilton–Jacobi-like equation for the generating function of the transformation—with the added
advantage that this equation becomes a set of simple algebraic equations for the unknown coefficients.
This strategy was first implemented by Langlois [3] for Friedmann–Lemaître–Robertson–Walker
(FLRW) cosmologies (see [4–7] for a recent discussion), and it was extended to anisotropic Bianchi I
spacetimes in [8] (see also [9,10] for a previous analysis starting from Einstein’s equations).

Although more systematic and geometric, the phase space derivation of linear cosmological
perturbations is still extremely tedious and lengthy, in particular in the presence of anisotropies.
In fact, the complexity of the calculations is one of the main barriers that researchers find when
entering this field. The goal of this paper is to alleviate this issue by introducing a computational
algorithm to find gauge invariant linear perturbations in Bianchi I spacetimes and the equations
of motion they satisfy in the symbolic language of Mathematica. Our algorithm is based on the
package xPert written for Mathematica, which is embedded on the tensor algebra packages xTensor
of the xAct distribution [11–13] (they are available under the General Public License). For the
application to perturbations on spherically symmetric spacetimes see also [14,15]. The merit of
our algorithm is that we combine a series of analytical and computational techniques to make the
problem tractable. We find that the phase space formulation enormously helps in these respects,
and when combined with carefully thought numerical algorithms, it gives rise to a computational
infrastructure of great utility to researchers in this field. This manuscript is accompanied by a
Mathematica notebook, publicly available at [16], that contains a step-by-step implementation of
the algorithms presented here. Furthermore, we have complemented the analysis with another code,
based on the C programming language and publicly available in [17], to solve the equations of motions
of gauge invariant perturbations and to compute observable quantities in the cosmic microwave
background (CMB) starting from suitable initial data, although the details of this code are not spelled
out in this paper.

This article is organized in such a way that the procedure has been separated into small steps
that are, one by one, introduced theoretically and implemented computationally. Namely, Section 2
introduces Bianchi I spacetimes and linear perturbations thereon in the Arnowitt–Deser–Misner
(ADM) formalism [2]. Section 3 introduces the generalization of the scalar-vector-tensor decomposition
commonly used in FLRW backgrounds. Section 4 introduces gauge invariant fields, and in Section 5,
we study their dynamics. Section 6 contains a short summary and some concluding remarks.

2. Perturbed Bianchi I Spacetimes in the ADM Formalism

2.1. Summary of the Theory

We start with Einstein’s gravity minimally coupled to a scalar field Φ with potential energy
density V(Φ) and no anisotropic stresses. We assume the spacetime manifold to be M = R×M3,
where M3 has the R3 topology. In the following, we will restrict ourselves to a finite volume V0 relative
to an auxiliary flat Euclidean metric δij defined in M3 (This is equivalent to putting the universe in
a box of arbitrarily large, but finite volume V0, with periodic boundary conditions. We do this for
convenience in the expressions below. The volume V0 will not impact predictions, and it can be taken
to infinity at the end of the calculations.). We will adopt a Hamiltonian formulation following ADM [2].
In this formalism, elements of the phase space are made of four real fields (Φ(~x), PΦ(~x), hij(~x), πij(~x))
defined in M3, where Latin indices i, j run from one to three. Here, hij(~x) is a Riemannian metric that
describes the intrinsic spatial geometry of M3 and πij(~x) its conjugate momentum. The non-vanishing
Poisson brackets between these fields are:

{Φ(~x), PΦ(~x′)} = δ(3)(~x−~x′) , {hij(~x), πkl(~x′)} = δk
(iδ

l
j)δ

(3)(~x−~x′) . (1)

where δk
(iδ

l
j) ≡

1
2 (δ

k
i δl

j + δk
j δl

i ). Dynamics is generated by the Hamiltonian:

H =
∫

d3x
[

N(~x) S(~x) + Ni(~x)Vi(~x)
]

, (2)

which is a combination of first class constraints, and N(~x) and Ni(~x) (the lapse and shift, respectively)
play the role of Lagrange multipliers. Concretely, S(~x) is the scalar constraint, and Vi(~x) are the vector
or diffeomorphism constraints. In terms of the canonical variables, they have the form:

S(~x) =
2κ√

h

(
πijπij −

1
2

π2
)
−
√

h
2κ

(3)R +
1

2
√

h
P2

Φ +
√

h V(Φ) +

√
h

2
DiΦDiΦ ≈ 0 , (3)

Vi(~x) = −2
√

h hij Dk(h−1/2πkj) + PΦ DiΦ ≈ 0 , (4)

where κ = 8πG, and Di is the covariant derivative compatible with the spatial metric hij, h its
determinant, and (3)R its Ricci scalar curvature. Given N(~x), Ni(~x), and a solution to Hamilton’s
equations, hij(~x, t), the spacetime metric takes the form:

ds2 = −(N2 − Ni Ni)dt2 + 2Ni dxidt + hij dxidxj, (5)

where t is a time variable that labels each space-like hyper-surface M3(t) and xi are spatial coordinates
on them.

Let us now focus on the sector of the phase space of general relativity that is made of Bianchi I
geometries together with small inhomogeneous perturbations. This is commonly done by considering
curves γ[ε] in the ADM phase space that pass through the Bianchi I subspace at ε = 0. Expanding the
phase space variables around ε = 0, we have:

hij(~x, ε) = h̊ij + ε δh(1)ij (~x) + . . . +
εn

n!
δh(n)ij (~x) + . . . ,

πij(~x, ε) = π̊ij + ε δπij (1)(~x) + . . . +
εn

n!
δπij (n)(~x) + . . . ,

Φ(~x, ε) = φ + ε δφ(1)(~x) + . . . +
εn

n!
δφ(n)(~x) + . . . , (6)

PΦ(~x, ε) = pφ + ε δp(1)φ (~x) + . . . +
εn

n!
δp(n)φ (~x) + . . . ,

where φ, pφ, h̊ij, π̊ij describe a Bianchi I background geometry, and δφ(n)(~x), δp(n)φ (~x), δh(n)ij (~x),

δπij (n)(~x) describe the nth-order perturbations thereon. The homogeneous variables satisfy the
Poisson brackets:

{φ, pφ} =
1
V0

, {h̊ij, π̊kl} = 1
V0

δk
(iδ

l
j) . (7)

As is common in the literature, we restrict ourselves to diagonal Bianchi I metrics, such that the
phase space variables h̊ij and π̊ij take a diagonal form in an appropriate system of coordinates xi:

h̊ij = diag(a2
1, a2

2, a2
3) , π̊ij = diag

(
πa1

2 a1
,

πa2

2 a2
,

πa3

2 a3

)
, (8)

where ai define the three directional scale factors; it follows from (7) that ai and πaj are canonically
conjugate, {ai, πaj} = 1

V0
δij (note that the subscripts i, j in ai and πaj are just labels, and not tensorial

indices). From now on, we will raise and lower all spatial indices i, j, k, ... with h̊ij and its inverse.
The next step is to expand the constraints (3) and (4) in perturbations:

S(~x) = S(0) + S(1)(~x) + S(2)(~x) + S(3)(~x) + · · · ,

Vi(~x) = V(0)
i +V(1)

i (~x) +V(2)
i (~x) +V(3)

i (~x) + · · · , (9)

where the superscripts in parenthesis denote the order in our perturbative expansion. As mentioned
before, we want to focus on linear perturbations. This will require, on the one hand, to keep only
first order perturbations δh(1)ij (~x), δπij (1)(~x), δφ(1)(~x), and δp(1)φ (~x) (since these will be the only

perturbations in the rest of this paper, from now on, we will remove the label (1)) and, on the
other hand, to truncate all constraints at quadratic order in these fields.

In addition to the perturbative expansion of the constraints, we also expand the lapse function as
N + δN(~x) and the shift as Ni + δNi(~x), where in the following, N is a homogeneous function, and we
take Ni = 0, so the background line element takes the familiar form:

ds2 = −N2dt2 + h̊ij dxidxj. (10)

In the rest of this section, we discuss the background sector and leave the study of the
inhomogeneous degrees of freedom for the next section. Because of homogeneity, V(0)

i identically
vanish (note that they are proportional to derivatives in space-like directions). Hence, the homogeneous
degrees of freedom are subject to only one constraint, S(0), which takes the form:

S(0) =
1

2
√

h̊

[
κ

(
a2

1π2
a1

2
+

a2
2π2

a2

2
+

a2
3π2

a3

2
− a1πa1 a2πa2 − a2πa2 a3πa3 − a3πa3 a1πa1

)

+p2
φ + 2h̊V(φ̄)

]
≈ 0, (11)

with h̊ = (a1a2a3)
2 = a6 the determinant of h̊ij, and we have defined a ≡ (a1a2a3)

1/3 as the mean scale
factor. Then, the Hamiltonian (2) reduces to:

HBI =
∫

M3

d3x N S(0) = V0 N S(0) . (12)

If we choose N = 1,HBI generates evolution in standard cosmic time t. Hamilton’s equations of
motion then read:

ȧi = {ai,HBI}, π̇ai = {πai ,HBI} , (13)

φ̇ = {φ,HBI} , ṗφ = {pφ,HBI} .

These equations fully determine the dynamical evolution of the Bianchi I geometry, once suitable
initial data satisfying the scalar constraint are provided. For convenience in the interpretation of the
solutions, it is useful to introduce (see, e.g., [8] for details), on the one hand, the average Hubble rate
H = ȧ

a . Its relation to the directional Hubble rates Hi ≡ ȧi
ai

is H = 1
3 (H1 + H2 + H3). On the other

hand, the anisotropic shear σij is defined from π̊ij by:

π̊ij =
h1/6

6
πa h̊ij +

h1/2

2κ
σij , (14)

where πa is the conjugate momenta of a (it can be written in terms of ȧ as πa = − 6
κ a ȧ). Equation (14)

is equivalent to saying that the components of σij:

σij = diag(a2
1 σ1, a2

2 σ2, a2
3 σ3) , (15)

are related to the canonical variables πai by:

πai =
1
κ

a3

ai
(σi − 2 H) . (16)

Using the equations of motion (13), one can check that σi = Hi − H, and from this, it is obvious
that σi are not all independent, but satisfy σ1 + σ2 + σ3 = 0—or in other words, σij is traceless. It is also
convenient to define the shear squared:

σ2 = σijσ
ij = σ2

1 + σ2
2 + σ2

3 = (H1 − H)2 + (H2 − H)2 + (H3 − H)2 . (17)

With these definitions, the equations of motion (13) can be written in the more familiar form:

ä
a
= −κ

6
[ρ + 3 P]− σ2

3
, φ̈ + 3

ȧ
a

φ̇ +
dV(φ)

dφ
= 0 , (18)

and the scalar constraint (11) as:

H2 =
κ

3
ρ +

σ2

6
. (19)

Here, ρ ≡ 1
2 φ̇2 +V(φ) and P ≡ 1

2 φ̇2−V(φ) are the energy and pressure densities of φ, respectively.
The expressions above are equivalent to the diagonal components of Einstein’s equations. For σ2 =

0, they reduce to the Friedmann–Lemaître theory of isotropic cosmologies. On the other hand,
the equations of motion for the shear are:

σ̇i
j = −3 H σi

j , (20)

whose solutions are simply σi = Σi/a3, where Σi are constants constrained by Σ1 + Σ2 + Σ3 = 0.
This implies σ2 = Σ2

a6 , with Σ2 ≡ Σ2
1 + Σ2

2 + Σ2
3.

2.2. Implementation in Mathematica

We will begin here the description of the main steps carried out in the Mathematica notebook [16].

2.2.1. Preliminaries

The perturbative expansions are carried out employing the package xPert [13]. Hence, the first
step is to load this package with the following command:

In[1] := «xAct‘xPert‘

We define the three-dimensional manifold M3 with abstract indices {i, j, k, l, m, n}:

In[2] := DefManifold[M3,3,{i, j, k, l, m, n}];

The spatial slices will be parameterized by a time variable t. We define it with the command:

In[3] := DefParameter[t,PrintAs->"t"];

We also define the gravitational coupling constant:

In[4] := DefConstantSymbol[κ];

The (Riemannian) spatial metric h and its covariant derivative CD are introduced by means of:

In[5] := DefMetric[1,h[-i,-j],CD,{";","D"}, Otherdependencies->{t},WeightedWithBasis->AIndex];

Note that we have allowed the spatial metric h to depend on the time parameter t. We now
introduce perturbations of the metric:

In[6] := DefMetricPerturbation[h,δh,ε];

We define the scalar field:

In[7] := DefTensor[φ[],{M3,t}];

and its potential:

In[8] := DefScalarFunction[V];

We incorporate perturbations of the scalar field with:

In[9] := DefTensorPerturbation[δφ[LI[1]],φ[],{M3,t}];

We define next canonical momenta. The momentum conjugate to the spatial metric is
introduced as:

In[10] := DefTensor[P[i,j],{M3,t}];

and its perturbations are defined by:

In[11] := DefTensorPerturbation[δP[LI[1],i,j],P[i,j],{M3,t}];

Finally, we define the momentum conjugate to the scalar field:

In[12] := DefTensor[Pφ[],{M3,t}];

and its corresponding perturbation:

In[13] := DefTensorPerturbation[δPφ[LI[1]],Pφ[],{M3,t}];

2.2.2. Scalar and Vector Constraints

We start introducing the diffeomorphism constraints defined in (4):

In[14] := diffeo=-2PD[-k]@(h[-i,-j]P[j,k])+P[k,j]PD[-i]@h[-k,-j]+Pφ[]PD[-i]@φ[];

As we will see below, only the linear term in the perturbative expansion of these constraints,
called V(1)

i (~x) above, will be relevant in the description of linearized perturbations (recall also that

V(0)
i are identically zero). They are defined in the notebook by:

In[15] := (Perturbed[diffeo,1]/ε);
diffeo1 = % /.MakeRule[PD[-i]@φ[],0]/.MakeRule[PD[-i]@h[-j,-k],0]
/.MakeRule[PD[-i]@P[j,k],0];

In this expression, we have imposed that the partial spatial derivatives of the background degrees
of freedom vanish because of homogeneity.

Let us focus now on the scalar constraint. We are going to compute each of its terms, written
in (3), separately. First of all, the three-dimensional Ricci curvature is:

In[16] := ricci=(Deth[])]∧(1/2)h[j,k]RiemannCD[-j,-i,-k,i]//RiemannToChristoffel
//ChristoffelToMetric//Simplification//NoScalar;

We now expand this term in perturbations by using:

In[17] := Perturbed[ricci,2]/.MakeRule[{PD[-i]@h[-j,-k],0}]//ExpandPerturbation;
r2 = %/.MakeRule[h[LI[2],-i,-j],0];

where we have imposed again homogeneity of the background metric, ∂ihjk = 0, and we have put to

zero the second order perturbations, δh(2)ij = 0.
On the other hand, the first term in (3) (the “kinetic” term of the gravitational sector) is:

In[18] := (Deth[])∧(-1/2)P[i,j]P[k,l](h[-i,-k]h[-j,-l]-h[-i,-j]h[-k,-l]/2);
ExpandPerturbation[Perturbed[%,2]];
pipi = % /.MakeRule[{δP[LI[2],-i,-j],0}]/.MakeRule[{δh[LI[2],-i,-j],0}];

where we have imposed δh(2)ij = 0 and δπ
(2)
ij = 0 in the last line.

The terms in (3) that depend on the scalar field are:

In[19] := 1/2Deth[]∧(-1/2)Pφ[]∧2+Deth[]∧(1/2)(1/2PD[-i]@φ[]PD[-j]@φ[]h[i,j]
+V[φ[]]);
ExpandPerturbation[Perturbed[%,2]];
matter = % /.MakeRule[{δh[LI[2],-i,-j],0}]/.MakeRule[{δφ[LI[2]],0}]
/.MakeRule[{δPφ[LI[2]],0}]/.MakeRule[{PD[-i]@φ[],0}];

Putting everything together, the scalar constraint, up to second order in perturbations, is:

In[20] := S = Series[(2κ)pipi-1/(2κ)r2+matter,{ε,0,2}];

The scalar constraint contributes with 0th, 1st and 2nd order terms in the perturbative expansion.
Let us identify each one. We first introduce the shear tensor as:

In[21] := DefTensor[σ[i,j],{M3,t},Symmetric[{i,j}]];

and the shear σ ≡
√

σ2

In[22] := DefTensor[σb[],{M3,t}];

We impose that the shear is traceless and symmetric, and its relation with σb[]∧2, with the
following automatic rules:

In[23] := AutomaticRules[σ,MakeRule[{σ[i,j]h[-i,-j],0}]];
AutomaticRules[σ,MakeRule[{σ[i,-i],0}]];
AutomaticRules[σ,MakeRule[{σ[i,j]σ[k,l]h[-i,-k]h[-j,-l],σb[]∧2}]];

In addition, the conjugate variable πa to the average scale factor a is defined as:

In[24] := DefTensor[πa[],{M3,t}];

Expression (14) above is implemented as:

In[25] := bgmomrule=MakeRule[{P[i,j],πa[]/6 Deth[]∧(1/6) h[i,j]+Deth[]∧(1/2)/(2κ)
σ[i,j]}];

With this, we express the first-order diffeomorphism constraints in terms of the shear as:

In[26] := diffeo1/.bgmomrule//org//ChristoffelToMetric//Simplification//NoScalar;
diffeoa=%/.MakeRule[{PD[-i]@h[-j,-k],0}];

Similarly, the zeroth order scalar constraint is:

In[27] := S0 = SeriesCoefficient[S,0];

We write it in terms of σ2 and πa by:

In[28] := S0a=S0/.bgmomrule//ToCanonical;

In a similar way, we define the part of the scalar constraint that is linear in perturbations as:

In[29] := S1a = SeriesCoefficient[S,1];

and in terms of shear:

In[30] := S1b = S1a/.bgmomrule//ToCanonical;

The part of the scalar constraint that is quadratic in perturbations will be discussed in Section 5.

3. Scalar-Vector-Tensor Decomposition

3.1. Summary of the Theory

Linear perturbations satisfy, via (1) and (7), the canonical Poisson brackets:

{δφ(~x), δpφ(~x′)} = δ(3)(~x−~x′)− 1
V0

; {δhij(~x), δπkl(~x′)} = δk
(iδ

l
j)

(
δ(3)(~x−~x′)− 1

V0

)
. (21)

For convenience, we Fourier expand these perturbations and their conjugate momenta
(The Fourier expansion of fields is adapted to the fiducial cell of volume V0, so the wavenumbers~k
will take values on a discrete lattice~k ∈ 2π/(V0)

1/3 Z3. In the limit V0 → ∞, one recovers~k ∈ R3).

δφ(~x) = ∑
~k 6=~0

δφ̃(~k) ei~k·~x ; δpφ(~x) = ∑
~k 6=~0

δ p̃φ(~k) ei~k·~x , (22)

δhij(~x) = ∑
~k 6=~0

δh̃ij(~k) ei~k·~x ; δπij(~x) = ∑
~k 6=~0

δπ̃ij(~k) ei~k·~x , (23)

where~k ·~x = ki xi and such that ki is time independent (the comoving wavevector).
The Poisson brackets (21) become:

{δφ̃(~k), δ p̃φ(~k′)} = V−1
0 δ~k,−~k′ ; {δh̃ij(~k), δπ̃kl(~k′)} = V−1

0 δk
(iδ

l
j) δ~k,−~k′ . (24)

We now perform a generalization of the scalar-vector-tensor decomposition of δh̃ij(~k) and δπ̃ij(~k)
that is commonly used in FLRW spacetimes. Although this decomposition is adapted to the rotational
invariance of FLRW geometries, it will also be useful in Bianchi I, since it will allow us to work with
variables that become the familiar scalar, vector, and tensor modes when the background geometry
isotropizes (as it quickly happens if there is a phase of inflation). We define now a basis of 3× 3
symmetric matrices as:

A(1)
ij =

h̊ij√
3

, A(4)
ij =

1√
2

(
k̂i ê2j + k̂ j ê2i

)
,

A(2)
ij =

√
3
2

(
k̂i k̂ j −

h̊ij

3

)
, A(5)

ij =
1√
2

(
ê1i ê1j − ê2i ê2j

)
, (25)

A(3)
ij =

1√
2

(
k̂i ê1j + k̂ j ê1i

)
, A(6)

ij =
1√
2

(
ê1i ê2j + ê1j ê2i

)
.

Here, k̂ is the unit vector (with respect to h̊ij) in the direction of~k, and ê1, ê2 are two unit vectors
orthogonal among themselves and to k̂ (Note that the three unit vectors k̂, ê1, ê2 are time dependent.
This is because, on the one hand, the norm of ki is time dependent and, on the other hand, the unit
vectors ê1, ê2 need to rotate in time to remain orthogonal to k̂, unless k̂ points in one of the principal
directions. For the details on how to compute the time dependence of the unit vectors k̂, ê1, and ê2,
see Appendix A in [8]). We now define γn(~k) and πn(~k) as the components of δh̃ij(~k) and δπ̃ij(~k),
respectively, in this basis:

δh̃ij(~k) =
6

∑
n=1

γn(~k) A(n)
ij (k̂) ; δπ̃ij(~k) =

6

∑
n=1

πn(~k) Aij
(n)(k̂) . (26)

In FLRW spacetimes, γn and πn are called scalar modes for n = 1, 2, vector modes for n = 3, 4,
and tensor modes for n = 5, 6, due to their properties under rotations around the direction k̂. We will
keep using these names throughout this paper. The non-zero Poisson brackets of these modes are:

{γn(~k), πm(~k′)} = V−1
0 δnm δ~k,−~k′ . (27)

Furthermore, we define:

γ0 ≡
√

4κ δφ̃(~k), π0 ≡
√

1/4κ δ p̃φ(~k), (28)

and we denote all the degrees of freedom in perturbations as γα(~k) and πα(~k) with α = 0, · · · , 6.
It will be useful in the next section to define the products of the shear tensor σij and Aij

(n) as:

σ(n)(k̂) ≡ σij Aij
(n)(k̂), (29)

for n = 2, · · · , 6 (σ(1) vanishes, because it is proportional to the trace of σij). Note however that σ(n)(k̂)
is not the Fourier transform of any of the components of σij.

We can now write the linear constraints S(1)(~x) and V(1)
i (~x) in terms of the variables γα(~k) and

πα(~k). For this purpose, we first expand the constraints in Fourier modes S(1)(~x) = ∑~k S̃
(1)(~k)ei~k·~x and

V(1)
i (~x) = ∑~k Ṽ

(1)
i (~k)ei~k·~x, and then, we replace (26). Explicit expressions are provided in Appendix B

of [8].

3.2. Implementation in Mathematica

We begin by defining the vectors~k, ê1, and ê2 as follows:

In[31] := DefTensor[kv[-i],{M3,t}];

In[32] := DefTensor[e1[-i],{M3,t},OrthogonalTo->kv[i]];

In[33] := DefTensor[e2[-i],{M3,t},OrthogonalTo->{kv[i],e1[i]}];

We define the norm of~k as:

In[34] := DefTensor[k[],{M3,t}];

In[35] := AutomaticRules[kv,MakeRule[{kv[-i]kv[-b]h[i,j],k[]∧2}]];

and we add automatic rules to indicate that ê1 and ê2 are unit vectors:

In[36] := AutomaticRules[e1,MakeRule[{e1[-i]e1[-b]h[i,j],1}]];

In[37] := AutomaticRules[e2,MakeRule[{e2[-i]e2[-b]h[i,j],1}]];

The scalar, vector, and tensor modes are defined as follows. First, the symmetric matrix A(1)
ij is

introduced as:

In[38] := DefTensor[A1[-i,-j],{M3,t},Symmetric[{-i,-j}]];

and γ1(~k) and π1(~k) as:

In[39] := DefTensor[γ1[],{M3,t}];

In[40] := DefTensor[π1[],{M3,t}];

In the same way, we define all other tensors A2[-i,-j], . . . , A6[-i,-j] and the modes γ2[], . . . ,
γ6[], and π2[], . . . , π6[].

We implement the traces and orthogonality properties of these matrices as automatic rules using
the command AutomaticRules. However, for convenience, we express the matrices A(n)

ij in terms of
the background metric h and the orthogonal vectors kv[-i], e1[-i], and e2[-i] with the command
MakeRule. For instance, for the matrices associated with scalar modes, we define:

In[41] := A1rule=MakeRule[{A1[-i,-j],h[-i,-j]/Sqrt[3]}];

In[42] := A2rule=MakeRule[{A2[-i,-j],Sqrt[3/2](kv[-i]kv[-j]/k[]∧2-h[-i,-j]/3)}];

and similarly for the matrices corresponding to vector and tensor modes. Finally, we define γ0(~k) and
π0(~k):

In[43] := DefTensor[γ0[],{M3,t}];

In[44] := DefTensor[π0[],{M3,t}];

and their relation with the perturbations of the scalar field and its momentum:

In[45] := γ0rule=MakeRule[{δφ[LI[1]],γ0[]/Sqrt[4κ]}];
π0rule=MakeRule[{δPφ[LI[1]],π0[]Sqrt[4κ]}];

We can now introduce the rules that implement the decomposition (26):

In[46] := moderule1=MakeRule[{δh[LI[1],-i,-j],γ1[]A1[-i,-j]+γ2[]A2[-i,-j]+γ3[]A3[-i,-j]
+γ4[]A4[-i,-j]+γ5[]A5[-i,-j]+γ6[]A6[-i,-j]}];
moderule2=MakeRule[δP[LI[1],i,j],π1[]A1[i,j]+π2[]A2[i,j]+π3[]A3[i,j]
+π4[]A4[i,j]+π5[]A5[i,j]+π6[]A6[i,j]];

Next, we implement the canonical Poisson brackets between γα(~k) and πα(~k) by means of the
following function:

In[47] := PoissonBracket[f_,g_,q_List,p_List]/;Length[q]==Length[p]:=D[f,{q}].D[g,{p}]
-D[f,{p}].D[g,{q}];
PoissonBracket[most__,q:Except[_List],p_]:=PoissonBracket[most,{q},p];
PoissonBracket[most__,p:Except[_List]]:=PoissonBracket[most,{p}];

If we introduce the following arrays:

In[48] := Q={γ0[],γ1[],γ2[],γ3[],γ4[],γ5[],γ6[]};
PQ={π0[],π1[],π2[],π3[],π4[],π5[],π6[]};

one can compute the Poisson brackets of any two phase space functions of perturbations. For instance:

In[49] := PoissonBracket[γ0[],π0[],Q,PQ];

Out[49] := 1

Below, we use this function to verify that the Poisson algebra of linear constraints is closed. It is
important to keep in mind that the conjugate variable to γα(~k) is πα(−~k). We will take this into account,
although we will not implement it explicitly in the notebook for the sake of simplicity.

We now define the components σ(n)(k̂) of the shear tensor following Equation (29):

In[50] := DefTensor[σ2[],{M3,t}];

and similarly for σ3[], . . ., σ6[]. The relation of these quantities and σij is implemented via the rule:

In[51] := sheardecomposition=MakeRule[{σ[-i,-j],σ2[]A2[-i,-j]+σ3[]A3[-i,-j]
+σ4[]A4[-i,-j]+σ5[]A5[-i,-j]+σ6[]A6[-i,-j]}];

It will be useful in the next sections to define the following rule:

In[52] := σ[-i,-j]σ[i,j]/.sheardecomposition//org;
σbrule=MakeRule[{σb[],%∧(1/2)}];

We finish this section by writing the linear constraints in Fourier space S̃(1)(~k) and Ṽ(1)
i (~k) and

in terms of the new variables γα(~k) and πα(~k). For the diffeomorphism constraints Ṽ(1)
i (~k), this is

implemented by applying the rules:

In[53] := diffeoa/.MakeRule[{PD[-b]@δh[LI[1],-i,-k],I*kv[-j]δh[LI[1],-i,-k]}]
/.MakeRule[{PD[-i]@δP[LI[1],j,k],I*kv[-d]δP[LI[1],j,k]}]
/.MakeRule[{PD[-i]@δφ[LI[1]],I*kv[-i]δφ[LI[1]]}]/.moderule1
/.moderule2/.γ0rule/.π0rule//ContractMetric;

We further simplify the final expression:

In[54] := diffeob = %/.sheardecomposition/.A1rule/.A2rule/.A3rule/.A4rule
/.A5rule/.A6rule//org;

Now, we decompose these constraints in their projections in the directions k̂, ê1, and ê2:

In[55] := diffeob1=kv[i]diffeob//ToCanonical;
diffeob2=e1[i]diffeob//ToCanonical;
diffeob3=e2[i]diffeob//ToCanonical;

We proceed in a similar way with S̃(1)(~k). We do it in several steps. In the first one, we substitute
∂i by i ki:

In[56] := S1b/.MakeRule[{PD[-i]@δh[LI[1],-j,-k],I*kv[-j]δh[LI[1],-i,-k]}]
/.MakeRule[{PD[-j]@δh[LI[1],-i,-k],I*kv[-j]δh[LI[1],-i,-k]}]
/.MakeRule[{PD[-i]@kv[i],0}]//ContractMetric;

We then apply the SVT decomposition by means of:

In[57] := %/.moderule1/.moderule2/.γ0rule/.π0rule//ToCanonical;

Finally, we simplify the result:

In[58] := S1c=%/.sheardecomposition/.A2rule/.A3rule/.A4rule/.A5rule/.A6rule
/.σbrule//org;

One can now check that these constraints have vanishing Poisson brackets (modulo the
background constraint). For instance,

In[59] := PoissonBracket[S1c,diffeob1,Q,PQ]/.bgconstraintrule//org

Out[59] := 0

where we have evaluated the background constraint on-shell. We have checked that the Poisson
brackets with the remaining constraints all vanish, in these cases identically (see the Mathematica
notebook [16]). Hence, they are first class constraints, as they must be from the view point of
general relativity.

One can also check that none of the modes γα or their conjugate momenta πα commute with these
constraints. We show here a couple of examples:

In[60] := PoissonBracket[γ0[],S1c,Q,PQ]//org

Out[60] := 2
√

κPφ√
˜̃h

In[61] := PoissonBracket[γ1[],S1c,Q,PQ]//org

Out[61] := − κπa√
3 ˜̃h

1/3

Consequently, these variables are not gauge invariant (this contrasts with the isotropic case, where
tensor modes γ5, γ6, and their momenta are gauge invariant).

4. Gauge Invariant Variables

4.1. Theory

We have seven degrees of freedom (per Fourier mode) in configuration variables in the
perturbations, γα(~k). They are subject to four first class constraints, S̃(1)(~k) ≈ 0 and Ṽ(1)

i (~k) ≈ 0,
that are the generators of gauge transformations for each mode. Each constraint reduces the number
of independent configuration variables by one. As a consequence, we are left with 7− 4 = 3 physical
configuration fields, together with their conjugate momenta. We will isolate these degrees of freedom
by identifying gauge invariant fields. In the Lagrangian formalism, this was accomplished in [9]. In
the Hamiltonian framework, the gauge invariant variables are defined as fields that are left invariant
by the gauge flow generated by the linear constraints, or equivalently that Poisson-commute with
them.

A conceptually simple and elegant procedure to find gauge invariant variables can be obtained
by using the ideas of [18], which were in part applied to FLRW cosmologies in [3]. The details of this
procedure in Bianchi I cosmologies can be found in [8]. In summary, the idea is to find a canonical
transformation from γα(~k), πα(~k) to new variables Γα(~k), Πα(~k) such that the new momenta Πα(~k) for
α = 3, 4, 5, 6 are proportional to the four constraints S̃(1)(~k) and Ṽ(1)

i (~k), respectively. More concretely,
we demand:

Π3(~k) = 1
|~k|

S̃(1)(~k) , Π4(~k) = 1
i |~k|

k̂j Ṽ(1)
j (~k) , Π5(~k) = 1

i |~k|
êj

1 Ṽ
(1)
j (~k) , Π6(~k) = 1

i |~k|
êj

2 Ṽ
(1)
j (~k) , (30)

where the factors 1/|~k|, with |~k| ≡
√

kiki 6= 0, have been introduced for dimensional reasons
and the imaginary unit for convenience. This choice is possible because the constraints are first
class, and it can be done globally in the perturbed phase space because of the linearity of the system.
If Equation (30) is satisfied, then the canonical commutation relations guarantee that Γα(~k) and
Πα(~k) for α = 0, 1, 2 Poisson-commute with the constraints, and hence, they are gauge invariant.
This procedure also guarantees that gauge invariant fields and pure gauge ones are decoupled in
the Hamiltonian (as we will see explicitly below), and hence, the dynamics does not mix them. One
can then consistently focus attention on gauge invariant perturbations. Interestingly, the task of
finding such a canonical transformation reduces to solving a Hamilton–Jacobi-like equation for a
generating function, and furthermore, by working in Fourier space, this equation reduces to algebraic
equations that are easy to solve in Mathematica. More concretely, we start with a generating function
in Fourier space:

G(~k) = Bαβ(~k)Πα(~k) γβ(~k) + Aαβ(~k) γα(~k)γβ(~k) , (31)

that we choose to be of type 2—i.e., it depends on old variables γα and new momenta Πα—and from
which the rest of the variables are given by:

πα(~k) =
∂G(γβ, Πβ)

∂γα(~k)
, Γα(~k) =

∂G(γβ, Πβ)

∂Πα(~k)
. (32)

where Bαβ and Aαβ are matrices whose components depend on background variables, but not on
perturbations, and furthermore, Aαβ is symmetric. Equations (32) provide then a set of algebraic
relations for the 77 unknown coefficients Bαβ and Aαβ, although only 38 of them are independent.
Hence, there is freedom in choosing gauge invariant variables. As mentioned above, we want to
choose gauge invariant fields that in the isotropic limit reduce to the familiar comoving curvature
perturbations and the two tensor modes. Indeed, gauge invariant fields Γα(~k) satisfying this property
can be identified by inspection, just by looking at the Poisson brackets of γα and the linear constraints.
They are:

Γ0(~k) = γ0 +

√
κ pφ√

1/6 κ a πa + a3 σ(2)

(√
2 γ1 − γ2

)
, (33)

Γ1(~k) = γ5 +
a2 σ(5)√

1/6 κ πa + a2 σ(2)

(√
2 γ1 − γ2

)
, (34)

Γ2(~k) = γ6 +
a2 σ(6)√

1/6 κ πa + a2 σ(2)

(√
2 γ1 − γ2

)
, (35)

This will be our choice of gauge invariant fields. Other choices are possible, and all of them can
be derived by using the companion notebook [16]. In the isotropic limit σ(n) → 0, Γ1 and Γ2 reduce to
the familiar two polarizations of transverse and traceless tensor modes, and Γ0 becomes proportional
to the comoving curvature perturbationR(~k) ≡ 1√

4κ
a
z Γ0, where z = − 6

κ
pφ

πa
= φ̇

H a.
Identifying the form of the new gauge invariant variables Γα in terms of γα will simplify the

computation of G(~k) (for instance, this determines the value of some of the coefficients Bαβ), but it is
important to emphasize that the method, and therefore its implementation in the algorithm reported in
our manuscript, allows us to work with gauge invariant variables and pure gauge ones in a systematic
way, without having to identify them a priori. To our knowledge, numerical tools meeting all these
requirements (i.e., handle efficiently complicated phase space functions and keep the construction
as general as possible) are not publicly available, at least in the context of cosmological perturbation
theory or similar settings.

In addition, we will demand (see the next section) the Hamiltonian to be “diagonal” in the
new configuration variables and momenta, i.e., we will eliminate “cross terms” of the type ΓαΠβ.
This aesthetic condition will impose further restriction in the coefficients Aαβ’s. The rest of free
coefficients can be equated to zero for simplicity. Once all the coefficients Aαβ and Bαβ are specified,
the form of the conjugate momenta Πα for α = 0, 1, 2 are obtained from (32).

4.2. Implementation in Mathematica

We summarize here the main steps of the code [16]. We start defining the new variables
Γα(~k), Πα(~k):

In[62] := DefTensor[Γ0[],{M3,t}];
DefTensor[Π0[],{M3,t}];

and similarly for α = 1, . . . , 6.
As explained above, rather than solving all the equations that constrain the coefficients Aαβ and

Bαβ, we simplify the calculation by identifying suitable gauge invariant variables (see Equation (35)
above). These variables are named in the code as Γ0new, Γ1new, Γ2new:

In[63] := Γ0new=γ0[]+(6 Sqrt[2 κ] Pφ γ1[])/((Deth[]∧(1/6)) (Sqrt[6] κ πa[]
+6 (Deth[]∧(1/6))∧2 σ2[]))-(6 Sqrt[κ] Pφ[] γ2[])/((Deth[]∧(1/6)) (Sqrt[6] κ πa[]
+6 (Deth[]∧(1/6))∧2 σ2[]));

and similarly for Γ1new, Γ2new. With this choice, we can determine some of the coefficients Bαβ.
The remaining coefficients in Bαβ, namely those with α, β = 3, 4, 5, 6, are written as unknowns CI[],
EI[], FI[], and JI[], with I={0, . . . , 6}. For example, we define:

In[64] := DefTensor[C0[],M3,t];
DefTensor[E0[],M3,t];
DefTensor[F0[],M3,t];
DefTensor[J0[],M3,t];

and similar definitions for the remaining unknowns. We use these coefficients to define:

In[65] := Γ3new=γ0[]C0[]+...+γ6[]C6[];
Γ4new=γ0[]E0[]+...+γ6[]E6[];
Γ5new=γ0[]F0[]+...+γ6[]F6[];
Γ6new=γ0[]J0[]+...+γ6[]J6[];

The coefficients Aαβ are denoted by AIJ[], with I,J={0, . . . , 6}. For instance, we define:

In[66] := DefTensor[A00[],M3,t];

and similarly for the remaining ones.
Using these coefficients and the definitions of new variables in terms of them, we introduce the

definition of the generating function as:

In[67] := G=Γ0newΠ0[]+Γ1newΠ1[]+. . .+Γ5newΠ5[]+Γ6newΠ6[]+A00[] γ0[] γ0[]
+2 A01[] γ0[] γ1[]+. . .+2 A56[] γ5[] γ6[]+A66[] γ6[] γ6[]//NoScalar;

We now follow the strategy described above. Namely, we first obtain an expression for the old
momenta πα in terms of γα and Πβ by taking the derivative of the generating function with respect to
γα. These expressions can then be substituted in the scalar and vector constraints, to express them in
terms of old configuration variables γα and new momenta Πβ. By equating these constraints to Πα for
α = 3, 4, 5, 6 as indicated in (30), we obtain 44 algebraic equations for the coefficients Aαβ, out of which
38 are independent. Some of the remaining coefficients can be solved by demanding that the new
Hamiltonian does not contain cross terms between new momenta and new configuration variables.
Finally, the remaining coefficients are set to zero for the sake of simplicity. See the notebook [16]
for details.

5. Dynamics of Gauge Invariant Perturbations

5.1. Theory

The dynamics of perturbations is generated by the second order scalar constraint
∫

d3x N S(2)(~x).
Let us notice that the second order vector constraints V(2)

i do not contribute since the homogeneous

part of the shift Ni is zero and the next contribution would come from δNi(~x)V
(2)
i , which is third

order in perturbations. If we expand the fields inside S(2)(~x) in Fourier modes, the expression for∫
d3x N S(2)(~x) can be written as ∑~k N S̃(2)(~k), where S̃(2)(~k) is quadratic in perturbations. In each

term, one perturbation is evaluated at~k and the other at −~k.
From the expression for S̃(2)(~k) in terms of δh̃ij and δπ̃ij, we obtain a Hamiltonian for the new

variables Γα(~k) and Πα(~k), by first implementing the change from δh̃ij(~k), δπ̃ij(~k) to γ(~k), π(~k) and
then from the latter to Γα(~k), Πα(~k). One has to keep in mind that these transformations involve
coefficients that depend on functions of the Bianchi I background geometry, and therefore, they have
to be understood as time-dependent quantities. This means that the final Hamiltonian is equal to the
original one in new variables, plus the time derivative of the generating function of the canonical

transformation, where the time derivative only affects the background functions. The first canonical
transformation can be implemented by the following generating function:

Gγ(~k) = −δπ̃ij(~k)
6

∑
n=1

A(n)
ij (~k)γn(~k) , (36)

where we have chosen it to be of the third type (i.e., it depends on new configuration variables and
old momenta). Recall that the matrices A(n)

ij (~k) depend on time. The second canonical transformation

from γα(~k), πα(~k) to Γα(~k), Πα(~k) is defined by the generating function (31).
After implementing these transformations, one can check that gauge invariant fields decouple

from pure gauge ones, and one obtains the following Hamiltonian for the former (see [8] for
further details):

Hpert =
N(t)V0

2 a(t) ∑
~k

2

∑
µ,µ′=0

[
4κ

a2(t)
δµ,µ′ |Πµ(~k)|2 +

a2(t)
4κ

(
δµ,µ′ k

2(t) + Uµµ′(t,~k)
)

Γµ(~k)Γ̄µ′(~k)
]

, (37)

where δµ,µ′ is the Kronecker delta and k2(t) ≡ a2(t) kik j = a2(t)
(

k2
1

a2
1(t)

+
k2

2
a2

2(t)
+

k2
3

a2
3(t)

)
. If we choose

N = 1, this Hamiltonian generates evolution in proper time t and in conformal time if N = a.
The (time-dependent) effective potentials Uµµ′ are symmetric in µ and µ′, and they become diagonal
(i.e., proportional to δµµ′ ) in the isotropic limit. However, in the presence of anisotropies, they couple
gauge invariant perturbations among themselves. They are explicitly written in Appendix A.

The equations of motion are (we use cosmic time):

Γ̇µ(~k) = {Γµ(~k),Hpert} =
4κ

a3 Πµ(~k) ,

Π̇µ(~k) = {Πµ(~k),Hpert} = −
a

4κ

2

∑
µ′=0

(δµµ′ k
2 + Uµµ′) Γµ′(~k) . (38)

Combining these equations into second order differential equations, we obtain:

Γ̈µ + 3 H Γ̇µ +
k2

a2 Γµ +
1
a2

2

∑
µ′=0
Uµµ′ Γµ′ = 0 . (39)

This is a set of three coupled, second order, ordinary differential equations for each wavevector
~k, and they reduce to the familiar (decoupled) equations for scalar and tensor perturbations in the
isotropic FLRW limit

5.2. Implementation in Mathematica

We start with the expression ∑~k N S̃(2)(~k) in terms of δh̃ij(~k) and δπ̃ij(~k). As mentioned above,
each term S̃(2)(~k) is quadratic in perturbations, with one field evaluated at~k and the other at −~k.
In our Mathematica notebook, this will remain implicit, since the code will be significantly simpler in
this way.

We first obtain an expression for S̃(2)(~k) in terms of δh̃ij(~k) and δπ̃ij(~k):

In[68] := S2a=SeriesCoefficient[S,2];

In[69] := S2a/.MakeRule[PD[-i]@PD[-j]@δh[LI[1],-k,-l],-kv[-i]kv[-j]δh[LI[1],-k,-l]]
/.MakeRule[PD[-i]@δφ[LI[1]]PD[-j]@δφ[LI[1]],kv[-i]kv[-j]δφ[LI[1]]δφ[LI[1]]]
/.MakeRule[PD[-i]@δh[LI[1],-j,-k] PD[-d]@δh[LI[1],-l,-m],kv[-i]kv[-l]
δh[LI[1],-j,-k]δh[LI[1],-l,-m]]/.MakeRule[h[i,j]kv[-i]kv[-j],k[]∧2]

/.MakeRule[PD[-j]@PD[-l]@δh[LI[2],-i,-k],0]
/.MakeRule[PD[-k]@PD[-l]@δh[LI[2],-i,-j],0]/.bgmomrule//org;

where, as before, we have replaced spatial derivatives by i~k. The second step is to move from
δφ̃(~k), δ p̃φ(~k), δh̃ij(~k), δπ̃ij(~k) to γα(~k), πβ(~k):

In[70] := %/.moderule1/.moderule2/.γ0rule/.π0rule//org;

and we also decompose the shear in its components σ(n):

In[71] := %/.sheardecomposition//org;

We further simply our final result with:

In[72] := S2b=%/.A2rule/.A3rule/.A4rule/.A5rule/.A6rule/.σbrule//org;

Finally, we want to write the Hamiltonian in terms of the new momenta Πα and configuration
variables Γα. In order to do so, we introduce in the notebook the expressions for old configuration
variables as functions of new ones as γ0old1, . . . , γ6old1. Similarly, we introduce expressions for old
momenta in terms of new momenta and old configuration variables, which are denoted by π0old1, . . .
, π6old1. One can easily combine these two sets of expressions to replace old configuration variables
and momenta in any expressions by new ones. Thus, we obtain:

In[73] := S2b/.MakeRule[π0[],π0old1]/.MakeRule[π1[],π1old1]/.MakeRule[π2[],π2old1]
/.MakeRule[π3[],π3old1]/.MakeRule[π4[],π4old1]/.MakeRule[π5[],π5old1]
/.MakeRule[π6[],π6old1];
S2c=%/.MakeRule[γ0[],γ0old1]/.MakeRule[γ1[],γ1old1]/.MakeRule[γ2[],γ2old1]
/.MakeRule[γ3[],γ3old1]/.MakeRule[γ4[],γ4old1]/.MakeRule[γ5[],γ5old1]
/.MakeRule[γ6[],γ6old1];

As explained above, the final Hamiltonian is obtained from this expression by adding the time
derivative of the generating functions (36) and (31). We denote by dGγdt1 and dGdt2 their time
derivatives in the companion notebook, respectively. The addition of these terms requires some
additional simplifications. Concretely, we apply the following rule in several intermediate steps:

In[74] := bgconstraintrule=MakeRule[V[φ[]],-1/(2κ)(σ2[]∧2+σ3[]∧2+σ4[]∧2+σ5[]∧2+σ6[]∧2)
-1/Deth[]∧(1/2)(Pφ[]∧2/(2Sqrt[Deth[]])-(κπa[]∧2)/(12Deth[]∧(1/6)))];

It uses the background constraint to replace the potential V(φ) of the scalar field by the other
background variables. We get:

In[75] := S2d = S2c+dGdt1+dGγdt2/.bgconstraintrule;

This expression can be further simplified. On the one hand, it contains terms proportional to the
new momenta Π3, . . . , Π6, which, by construction, are constrained to vanish. Therefore, we set them
equal to zero with the rule:

In[76] := S2e=S2d/.MakeRule[Π3[],0]/.MakeRule[Π4[],0]/.MakeRule[Π5[],0]/.MakeRule[Π6[],0]
//org;

We have verified that there is no coupling between gauge invariant and pure gauge variables,
once the background constraint is also imposed. Therefore, we can focus on the part of the Hamiltonian
that contains only gauge-invariant fields. Nevertheless, this part still contains cross terms between Πµ

and Γµ, with µ = 0, 1, 2. Fortunately, and as discussed above, the generating function G(~k) still has
some free parameters that can be fixed by requiring that these terms vanish. Concretely, we require the
coefficients multiplying the cross terms (Γ0[] Π0[]), (Γ1[] Π0[]), (Γ2[] Π0[]), (Γ1[] Π1[]),
(Γ1[] Π2[]), and (Γ2[] Π2[]), to vanish. These conditions fix the coefficients A01[], A05[],

A06[], A56[], A12[], A26[] in (31). As a bonus, this automatically guarantees that the remaining
cross terms (Γ0[] Π1[]), (Γ1[] Π0[]), (Γ0[] Π2[]), (Γ2[] Π0[]), (Γ1[] Π2[]), and (Γ2[]
Π1[]) all vanish. As mentioned before, we set the remaining free parameters to zero, in order to have
a Hamiltonian as simple as possible.

Finally, in order to focus on gauge invariant fields, we set pure gauge variables to zero in the code.
This yields an expression for the Hamiltonian that we denote by H2f, which is further simplified by
means of the rule:

In[77] := H2g=H2f/.bgconstraintrule//org;

This final Hamiltonian can be simplified and written in the form given in (37). Equations of
motion for the gauge invariant variables can be derived straightforwardly from it.

6. Discussion

We have described in this paper the main steps of a computer code written in the symbolic
language of Mathematica and made publicly available in [16] that derives gauge invariant linear
perturbations in Bianchi type I cosmological spacetimes in a Hamiltonian or phase space approach.
In this formulation, gauge invariant linear perturbations are defined by fields that Poisson-commute
with the linear constraints of the theory, and they can be found systematically by solving a canonical
transformation that identifies some of the new momenta with the constraints. We described in detail
the implementation of this procedure in Mathematica. Our code provides an efficient tool to explore
different choices of linear gauge invariant fields and to derive the equations of motion they satisfy.
It can also be used to work with gauge dependent variables, after choosing a gauge, and to relate
physical observables written in different gauges. Furthermore, we complemented this analysis with
a computer code, based on the C programing language, available in [17], that solves the equations
of motions and computes observables that can be compared with current and future data from the
cosmic microwave background (CMB). Our computer codes should be of great utility for researchers
interested in cosmological perturbations in Bianchi spacetimes and their consequences for the CMB, as
well as in isotropic FLRW, which is obtained by simply putting the anisotropies to zero. Our theoretical
and numerical analysis can also be of interest for pedagogical purposes, since it provides a step-by-step,
guided way of implementing the theory of gauge invariant cosmological perturbations in a computer.

Appendix A. Hamiltonian of Gauge Invariant Perturbations

The potentials Uµµ′ in the Hamiltonian (37) that generates the dynamics of gauge invariant
perturbations are given by the following expressions:

U00 = a2 Vφφ −
2κ p2

φF2

a3 + 2κF1

(
−

κ p2
φ pa

3a5 + 2 Vφ pφ

)
,

U01 = U10 = 2
√

κ
a2

(
−a2 pφ σ(5) F2 + a5Vφ σ(5) F1 − a2 pφ G5 F1 + κ

6 pφ pa σ(5) F1

)
,

U02 = U20 = 2
√

κ
a2

(
−a2 pφ σ(6) F2 + a5 Vφ σ(6) F1 − a2 pφ G6 F1 + κ

6 pφ pa σ(6) F1

)
,

U12 = U21 = 2 σ(5) σ(6)
(
a2 − a3 F2 + 2

3 κ a pa F1
)
−
(

2 a3 σ(6) G5 + 2 a3 σ(5) G6

)
F1

U22 = − 2 a2 σ2
(5) +

κpa σ(2)√
6
− a2

√
2
3G2 + 4

3 κ a pa σ2
(6) F1 − 4 a3 σ(6) F1 G6 − 2 a3 σ2

(6) F2,

(A1)

with Vφ ≡ dV/dφ, Vφφ ≡ d2V/dφ2, and:

F1 =
− κpa

2a3 +
√

3
2

σ(2)
a

2κρ+ σ2
(3) + σ2

(4)+ σ2
(5) + σ2

(6)
,

F2 =

3κ V
a −

κ2 p2
a

3a5 +
κpaσ(2)
2
√

6a3 +
√

3
2
G2
a −F1

[
κ2 p2

φ pa

a8 + 2 σ(3) G3 + 2 σ(4) G4+ 2 σ(5) G5 + 2 σ(6) G6)

]
2κρ+ σ2

(3) + σ2
(4)+ σ2

(5) + σ2
(6)

,

G2 =
κpaσ(2)

2 a2 −
√

3
2

(
σ2
(3) + σ2

(4)

)
,

G3 =
κ pa σ(3)

2 a2 + 1√
2

(√
3σ(2)σ(3) − σ(3)σ(5) − σ(4)σ(6)

)
,

G4 =
κpaσ(4)

2 a2 + 1√
2

(√
3σ(2)σ(4) + σ(4)σ(5) − σ(3)σ(6)

)
,

G5 =
κpaσ(5)

2 a2 + 1√
2
(σ2

(3) − σ2
(4)),

G6 =
κpaσ(6)

2 a2 +
√

2 σ(3)σ(4).

(A2)

Note that the expressions above have an implicit dependence on~k coming from σ(n)(~k).

